
Abstract
The problem of disjoint sets, also known asunion-find, consists
in maintaining a partition of a finite set within a data structure.
This structure provides two operations: a functionfind returning
the class of an element and a functionunionmerging two classes.
An optimal and imperative solution is known since 1975. However,
the imperative nature of this data structure may be a drawback
when it is used in a backtracking algorithm. This paper details the
implementation of a persistent union-find data structure asefficient
as its imperative counterpart. To achieve this result, our solution
makes heavy use of imperative features and thus it is a significant
example of a data structure whose side effects are safely hidden
behind a persistent interface. To strengthen this last claim, we
also detail a formalization using the Coq proof assistant which
shows both the correctness of our solution and its observational
persistence.

Categories and Subject Descriptors D.3.3 [Language Constructs
and Features]: Data types and structures; D.2.4 [Software/Program
Verification]: Correctness proofs

General Terms Algorithms, Verification

Keywords Union-Find, Persistence, Formal Verification

1. Introduction
The problem of disjoint sets, known asunion-find, consists in main-
taining a partition of a finite set within a data structure. Without loss
of generality, we can assume that we are considering a partition of
the n integers{0, 1, . . . , n − 1}. A typical signature1 for such a
data structure could be the following:

module type ImperativeUnionFind = sig
type t
val create : int → t
val find : t → int → int
val union : t → int → int → unit

end

It provides an abstract data typet for the partition and three opera-
tions. Thecreate operation takes an integern as input and builds

1 This article is illustrated with OBJECTIVECAML [2] code but is obviously
translatable to any other programming language.

a new partition where each element in{0, . . . , n− 1} constitutes a
class by itself. Thefind operation returns the class of an element,
as an integer considered to be the representative of this class. Fi-
nally theunion operation merges two classes of the partition, the
data structure being modified in place (hence the return typeunit).

An optimal solution is known since the 70s. It is attributed
to McIlroy and Morris [3] and its complexity was analyzed by
Tarjan [17]2. It is a classic of the algorithmic literature (see for
instance [11] chapter 22). The code of this solution is givenin
appendix. The main idea is to link together the elements of each
class. For this purpose, the arrayparent maps each integeri to
another integer of the same class. Within each class, these links
form an acyclic graph where any path leads to the representative,
which is the sole element mapped to itself. The following figure
illustrates a situation where the set{0, 1, . . . , 7} is partitioned into
two classes whose representatives are 3 and 4 respectively.

Thefind operation simply follows the links until it finds the rep-
resentative. Theunion operation first finds out the two representa-
tives of the given elements and then links one to the other. Then two
key improvements are used to reach optimal efficiency. The first
one is calledpath compression: any element encountered during
the search performed byfind is re-mapped to the representative.
The second one consists in keeping an approximation of the size of
each class, called arank, and to choose for the representative of a
union the one with the largest rank.

The union-find data structure is used in many different contexts.
In most cases, the imperative implementation above is perfectly
suited — and even optimal. But as soon as we need to roll back to
previous versions of the union-find structure, the imperative nature
of this solution is a nuisance. Indeed, providing either an efficient
“copy” or “undo” operation turns out to be very difficult, mostly
because of path compression. Here are typical examples where
union-find is used together with backtracking:

• Unification: A typical use of a union-find data structure is uni-
fication. For instance, unification is massively used in Prolog
where backtracking is precisely a fundamental concept. Unifi-
cation is also used in ML type inference. In that case, this is
not a backtracking algorithm but, within an interactive toplevel,

2 In the following, we will abusively refer to this solution asTarjan’s algo-
rithm.



destructive unifications of non-generalized type variables must
be undone when typing errors are encountered.

• Decision procedures: Another example of a union-find data
structure used within a backtracking algorithm are decision
procedures where the treatment of equality is usually based
on such a structure and where the boolean part of formulas is
mostly processed using backtracking [13, 16].

A standard solution to the issue of backtracking in data struc-
tures is to turn topersistentdata structures (as opposed toephemeral
data structures) [12]. In this case, theunion operation now returns
a new partition and leaves the previous one unchanged. The signa-
ture of such a persistent data structure could be the following:

module type PersistentUnionFind = sig
type t
val create : int → t
val find : t → int → int
val union : t → int → int → t

end

It simply differs from signatureImperativeUnionFind in the re-
turn type ofunion, which is nowt instead ofunit. One easy
way to implement such a signature is to use purely applicative data
structures. An immediate — and probably widely used — solution
is to switch from an imperative array to a purely applicativemap.
Assuming a moduleM implementing maps from integers to inte-
gers, as follows

module M : sig
type t
val empty : t
val add : int → int → t → t
val find : int → t → int

end

we can directly implement the persistent union-find data structure
as a map, that is

type t = M.t

Thenfind andunion operations are easily implemented:

let rec find m i =
try find (M.find i m) with Not_found → i

let union m i j =
let ri = find m i in
let rj = find m j in
if ri <> rj then M.add ri rj m else m

Unfortunately, path compression cannot be implemented without
changing the type offind: one would needfind to return a new
partition but this would require the programmer to modify client
code. For this reason, such solutions are far less efficient than the
imperative implementation.

The main contribution of this paper is the design of a persistent
union-find data structure as efficient as the imperative implementa-
tion. To achieve this result, we consider a data structure with side
effects yet with signaturePersistentUnionFind, so that client
code may use it as if it was a purely applicative data structure. This
is thus an example of an imperative data structure whose sideef-
fects aresafely hiddenbehind a persistent interface. Another con-
tribution of this paper is to strengthen this last claim witha formal
proof. Indeed, we used the Coq proof assistant [1, 6] to show both
the correctness of our solution and its observational persistence.

There are known techniques to design persistent data struc-
tures based on purely applicative programming [14] but there is
no efficient solution to the union-find problem in this framework.

There are also techniques to make imperative data structures per-
sistent [12] but, again, they were not applied to the union-find data
structure. Moreover, no formal proof of the observational persis-
tence of an imperative data structure has ever been done to our
knowledge.

This paper is organized as follows. Section 2 details a persistent
union-find data structure which is almost as efficient as its imper-
ative counterpart. Then Section 3 analyzes the efficiency ofthis
solution, comparing it to several other implementations. Finally,
Section 4 details the formal proof.

2. A Simple Yet Efficient Solution
Our solution to the persistent union-find problem is simultaneously
very simple and very efficient. It mostly consists in keepingclose
to the original algorithm and its path compression, but substituting
persistent arraysin place of the usual arrays.

2.1 Persistent Arrays

A persistent array is a data structure providing the same operations
as a usual array, namely the manipulation of elements indexed
from 0 ton − 1, with low cost access and update operations, but
where the update operation returns a new persistent array and leaves
the previous one unchanged. A minimal signature for polymorphic
persistent arrays is:

module type PersistentArray = sig
type α t
val init : int → (int → α) → α t
val get : α t → int → α
val set : α t → int → α → α t

end

It is easy to implement this signature using purely applicative
dictionaries, such as balanced binary search trees from OCAML ’s
standard library. But as demonstrated later by our benchmarks, the
logarithmic cost of the access and update operations on suchstruc-
tures is already prohibitive. Fortunately, it is possible to achieve far
better efficiency for persistent arrays, as we will show later (Sec-
tion 2.3). Meanwhile, we are presenting a persistent version of Tar-
jan’s algorithm, independently of the implementation of persistent
arrays.

2.2 A Persistent Version of Tarjan’s Union-Find Algorithm

To be independent of persistent arrays, we naturally introduce a
parameterized module, a so-calledfunctor:

module Make(A : PersistentArray)
: PersistentUnionFind

= struct

As in the imperative version, the union-find structure is a pair
of arrays (rank containing the ranks of the representatives and
parent containing the links) but here we use persistent arrays:

type t = {
rank: int A.t;
mutable parent: int A.t

}

The mutable nature of the second field will be exploited to perform
path compression. The creation of a new union-find structureis
immediate:

let create n = {
rank = A.init n (fun → 0);
parent = A.init n (fun i → i)

}



To implement path compression, thefind function must perform
modifications on theparent array on its way back, once the rep-
resentative has been found. Here this array is persistent and thus
a new array must be built. For this purpose, we introduce an aux-
iliary function find aux returning both the representative and the
new version of theparent array:

let rec find_aux f i =
let fi = A.get f i in
if fi == i then
f, i

else
let f, r = find_aux f fi in
let f = A.set f i r in
f, r

Then we can define a functionfind which callsfind aux and
then modifies theparent field with a side effect, to allow path
compression in later accesses of this data structure:

let find h x =
let f,cx = find_aux h.parent x in
h.parent ← f;
cx

As we can notice, this function indeed has the expected type,that
is it returns a single integer, while performing path compression.
As in the imperative code, the data structure has been mutated by a
side effect but the set of representatives remains unchanged.

To implement theunion function, we still follow the imperative
code but we return a new data structure, namely a new record of
typet. It contains the new versions of therank andparent arrays:

let union h x y =
let cx = find h x in
let cy = find h y in
if cx != cy then begin
let rx = A.get h.rank cx in
let ry = A.get h.rank cy in
if rx > ry then
{ h with parent = A.set h.parent cy cx }

else if rx < ry then
{ h with parent = A.set h.parent cx cy }

else
{ rank = A.set h.rank cx (rx + 1);
parent = A.set h.parent cy cx }

end else
h

We finally get a code which is not longer than its imperative coun-
terpart. The most difficult task remains: to feed this functor with an
efficient implementation of persistent arrays.

2.3 Efficient Implementation of Persistent Arrays

An efficient solution to the problem of persistent arrays hasactually
been known for a long time. It seems that it is due to H. Baker [4]
who was using it to implement closures in a Lisp runtime.

2.3.1 Main Idea

The base idea is to use a usual array3 for the last version of the
persistent array and indirections for previous versions. For this
purpose, we introduce the following mutually recursive data types:

type α t = α data ref
and α data =

| Arr of α array

3 From now on, we use the term “array” for a usual array, thus modified in
place, and the term “persistent array” otherwise.

| Diff of int × α × α t

The typeα t is the type of persistent arrays. It is a reference on a
value of typeα data which indicates its nature: either an imme-
diate valueArr a with an arraya, or an indirectionDiff(i, v, t)
standing for a persistent array which is identical to the persistent
arrayt everywhere except at indexi where it containsv. The ref-
erence may seem superfluous but it is actually crucial. Creating a
new persistent array is immediate:

let init n f = ref (Arr (Array.init n f))

The access functionget is also straightforward. Either the persis-
tent array is an immediate array, or we need to consider the indirec-
tion and possibly to recursively access another persistentarray:

let rec get t i = match !t with
| Arr a →

a.(i)
| Diff (j, v, t’) →

if i == j then v else get t’ i

All the subtlety is concentrated into theset function. The idea
is to keep the efficiency of a usual array on the very last version
of the persistent array, while possibly decreasing the efficiency of
previous versions. When updating the persistent arrayt, there are
two possible cases:

• either t is a reference to an object of shapeArr a; in that
case, we are going to replacet with an indirection (which is
possible since it is a reference and not a value of typeα data),
modify the arraya in place and return a new reference pointing
to Arr a.

• or t is already an indirection, that is pointing to aDiff node;
then we simply create and return a new indirection.

This is done in the following code:

let set t i v = match !t with
| Arr a as n →

let old = a.(i) in
a.(i) ← v;
let res = ref n in
t := Diff (i, old, res);
res

| Diff →
ref (Diff (i, v, t))

As we can notice, a value of the shapeArr a can only be created
by theinit function. Thus a sequence of updates only allocates
a single array and an additional space which is proportionalto the
number of updates (sinceset clearly runs inO(1) space and time).
If we consider the following definition of four persistent arraysa0,
a1, a2 anda3

let a0 = create 7 0
let a1 = set a0 1 7
let a2 = set a1 2 8
let a3 = set a1 2 9

then the situation right after these declarations is illustrated in Fig-
ure 1, where each reference is displayed as a circle and aDiff
block by a labelled edge. Generally speaking, we have the follow-
ing invariant: the graph of references of typeα t for the various
versions of a persistent array is acyclic and from any of these refer-
ences there is a unique path to theArr node.

Such persistent arrays achieve good results when we always
access the last version but efficiency greatly decreases when we
access previous versions. Indeed, a sequence of several updates
creates a linked list ofDiff nodes and then any access from this
list has a cost which is proportional to its length, that is tothe



Figure 1. Illustrating persistent arrays (1/2)

number of updates. This can be dramatic in a context where we
need to backtrack. But this is precisely the reason why we were
building persistent arrays. Fortunately, there exists a very simple
way to improve this first implementation.

2.3.2 A Major Improvement

To lower the cost of operations on previous versions of persistent
arrays, H. Baker introduces a very simple improvement [5]: as soon
as we try to access a persistent array which is not an immediate
array we first reverse the linked list leading to theArr node, to
move it in front of the list, that is precisely where we want to
access. This operation, that Baker callsrerooting, can be coded by
the following reroot function which takes a persistent array as
argument and returns nothing; it simply modifies the structure of
pointers, without modifying the contents of the persistentarrays.

let rec reroot t = match !t with
| Arr → ()
| Diff (i, v, t’) →

reroot t’;
begin match !t’ with
| Arr a as n →

let v’ = a.(i) in
a.(i) ← v;
t := n;
t’ := Diff (i, v’, t)

| Diff → assert false
end

After calling this function, we have the property thatt now points
to a value of the shapeArr. Thus we can modify the access function
so that it now calls thereroot function as soon as the persistent
array is aDiff node:

let rec get t i = match !t with
| Arr a →

a.(i)
| Diff →

reroot t;
begin match !t with
| Arr a → a.(i)
| Diff → assert false

end

We can modify theset function in a similar way:

let set t i v =
reroot t;
match !t with
| Arr a as n → ... as previously ...
| Diff → assert false

Going back to the situation of Figure 1, let us assume that we
now try to accessa1. Thenreroot is going to be called ona1. This
results ina1 now pointing to theArr node anda2 pointing to an
indirection, the values 0 and 8 at index 2 being swapped between

Figure 2. Illustrating persistent arrays (2/2)

the array and the indirection. This new situation is illustrated in
Figure 2.

The reroot operation has a cost proportional to the number
of Diff nodes that must be followed to reach theArr node, but
it is only performed the first time we access to an old version of
a persistent array. Any subsequent access will be performedin
constant time. To put it otherwise, we pay only once the cost of
coming back to a previous version. In a backtracking context, this
is a perfect solution. If the number of array operations is far greater
than the number of backtracks, then the amortized complexity of
theget andset operations will beO(1) in space and time.

It is important to notice, however, that if we manipulate simulta-
neously several versions of a single persistent array then efficiency
will decrease since thereroot function is going to spend much
time in list reversals.

Finally, we can also notice that thereroot function is not
tail-recursive. This can be an issue when we manipulate persistent
arrays on which many updates have been performed. It is however
easy to solve this issue by rewriting the code into continuation
passing style (CPS), without any real loss of efficiency. Thetests
which are presented later are performed using such a CPS version.

2.3.3 Final Improvements

It is possible to further improve this implementation. The first idea
is to optimize a call toset t i v when i is already mapped tov
in t. Then we save a useless indirection, and thus the allocationof
two blocks. This is especially efficient in the context of theunion-
find data structure, since path compression quickly maps allthe
elements to the representative and then subsequentset operations
become useless. (We could equivalently unroll thefind aux func-
tion and make a special treatment for paths of length 1.)

The second idea is to notice that in a context where weonly
perform backtracking, it is useless to maintain the contents of
persistent arrays that become unreachable when we go back toa
previous point. Indeed, these values are going to be immediately
reclaimed by the garbage collector. Thus we can still improve the
efficiency of our persistent arrays, in the particular case where we
are using them, that is where we go back to a previous versiont of
a persistent array without keeping any pointer to younger versions
of t.

The first modification is to introduce anInvalid node denoting
a persistent array where it is no more possible to access:

type α t = α data ref
and α data =
| Arr of int array
| Diff of int × α × α t
| Invalid

Then we modify thereroot function so that it does not reverse the
list of pointers but simply updates the contents of the array:

let rec reroot t = match !t with
| Arr → ()
| Diff (i, v, t’) →



reroot t’;
begin match !t’ with

| Arr a as n →
a.(i) ← v;
t := n;
t’ := Invalid

| Diff | Invalid → assert false
end

| Invalid → assert false

As we can notice, we save the allocation of aDiff node. The
remaining of the code is unchanged but is adapted to fail if wetry
to access to anInvalid node.

It is striking to notice that the final data structure we get isac-
tually nothing more than a usual array together with an undo stack,
that is the backtracking design pattern of the imperative program-
mer. But contrary to an imperative programming style where the
stack is made explicit inside the main algorithm, it is here hidden
behind an abstract data type which creates theillusion of persis-
tence.

3. Performance
We tried to test the efficiency of our solution in a situation as
realistic as possible. For this purpose we looked at the use made
by the Ergo decision procedure [8] of its internal union-finddata
structure. We distinguish three parameters:

• the number of backtracks;

• the total number offind andunion operations between two
branchings, denotedN ;

• the proportion ofunion operations with respect tofind oper-
ations, denotedp.

On the tests we made with the decision procedure, it happens
that the number of backtracks is small and that the proportion
of union operations is also quite small. Thus we chose to make
some tests following the branchings of a full binary search of
height 4. Between each node and its two sons we perform exactly
N operations. This traversal is executed forN = 20000, N =
100000 andN = 500000, with proportionp of union being equal
to 5, 10 and 15%.

Figure 3 displays the timings4 for various implementations. The
first line corresponds to the imperative implementation, asgiven in
appendix. It is incorrectly used here — we keep the current version
when backtracking —- but it is shown as a reference. The next four
lines are the successive refinements of our solution: the first version
of Section 2.3.2, the two improvements of Section 2.3.3 and finally
a manually defunctorized version of the final solution (defun.). The
last two lines are purely applicative solutions, in a purpose of com-
parison: the first (naı̈ve) is an AVL-based implementation without
path compression nor ranks, as described in the introduction, and
the second is the application of the functor from Section 2.2to per-
sistent arrays implemented as AVLs.

The results show that our final solution (defun.) is almost as
efficient as the imperative implementation. Note that the incorrect
use of the imperative code makes path compression on one branch
to be effective on all successive branches, resulting in maximal
path compression eventually. Results for the naı̈ve implementation
shows that it does not scale at all. The last line emphasizes that path
compression and ranking alone are not sufficient to achieve good
results, but that an efficient implementation of persistentarrays is
mandatory.

4 Timings are CPU time measured in seconds on a Pentium IV 2.4 GHz
running Linux.

4. A Formal Proof of Correctness
Even if our solution is conceptually simple, the data structures
are somewhat complex due to the massive use of (hidden) side
effects, in both persistent arrays and the persistent implementation
of Tarjan’s algorithm. For this reason, we decided to formally
verify the correctness of these two data structures and to show their
observational persistence. This section gives an overviewof this
formalization, conducted in the Coq proof assistant [1]. The whole
development can be found online5.

Section 4.1 briefly presents program verification using Coq,in-
troducing the notations used in the remainder. Section 4.2 describes
our Coq modeling of ML references. Then Section 4.3 presentsthe
verification of persistent arrays and Section 4.4 that of thepersis-
tent union-find data structure.

4.1 Program Verification using Coq

Coq is an interactive proof assistant based on the Calculus of
Inductive Constructions, a higher-order logic with polymorphism,
dependent types and a primitive notion of inductive types [10, 15].
In particular, this logic contains the purely applicative fragment
of ML and thus can be used to to define ML programs and to
show their correctness. Here is a short example on Peano natural
numbers. First, we define the datatypenat with two constructorsO
andS as we would do in ML:

Inductive nat : Set :=
| O : nat
| S : nat → nat

nat has typeSetwhich is the sort of datatypes. Predicates may also
be defined inductively. Here is the definition of a unary predicate
even on natural numbers:

Inductive even : nat → Prop :=
| evenO : even O
| evenSS : ∀n:nat, even n → even (S (S n)).

even has typenat → Prop whereProp is the sort of proposi-
tions. Such an inductive predicate is equivalent to the following
inference rules:

even O
(evenO)

even n

even (S (S n))
(evenSS)

ML-like functions are defined in a straightforward way:

Definition plus3 : nat → nat :=
fun n:nat ⇒ S (S (S n)).

Then it is possible to state properties about functions, such as “for
any even natural numbern, the result of(plus3 n) is not even”:

Lemma even_plus3 :
∀n:nat, even n → ~(even (plus3 n)).

Such a lemma declaration must be followed by a proof, which isa
list of tactic invocations. Proofs are omitted in this paper.

The richness of the Coq type system actually allows the user to
combine a function definition together with its correctnessproof.
To do so, the function is given a type which contains both types
for its arguments and result and a specification. Here is sucha
definition-as-proofof a functionf taking an even numbern as
argument and returning a resultm greater thann and even:

Definition f :
∀n:nat, even n → { m:nat | m > n ∧ even m }.

The type{ x : T | P (x) } is a dependent pair of a valuex of type
T and a proof ofP (x). The definition body forf is omitted and

5http://www.lri.fr/∼filliatr/puf/



p 5% 5% 5% 10% 10% 10% 15% 15% 15%
N 20000 100000 500000 20000 100000 500000 20000 100000 500000

Tarjan 0.31 2.23 12.50 0.33 2.34 12.90 0.34 2.36 13.20
2.3.2 0.52 3.03 17.10 0.81 4.78 26.80 1.16 6.78 37.90

2.3.3a 0.36 2.08 12.30 0.46 2.76 15.90 0.64 3.58 20.70
2.3.3b 0.34 2.01 11.70 0.42 2.54 14.90 0.52 3.21 18.70
defun. 0.33 1.90 11.30 0.41 2.45 14.40 0.52 3.14 17.80
naı̈ve 0.76 5.28 37.50 1.22 9.14 63.80 40.40 >10mn >10mn
maps 1.52 10.60 67.90 2.45 17.50 116.00 3.42 24.70 167.00

Figure 3. Performances

replaced by a proof script, which is both the function definition and
its correctness proof. Once the proof is completed, a mechanism
allows the user to extract an ML program from the definition-as-
proof.

Coq is naturally suited for the verification of purely applicative
programs. To deal with imperative programs, we need to modelthe
memory and to interpret imperative features as memory transform-
ers. This is similar to the formalization of operational semantics.

4.2 Modelling ML References

To model references, we introduce an abstract typepointer for
the values of references:

Parameter pointer : Set.

Then the set of all (possibly aliased) references of a given type is
modelled as a dictionary mapping each reference to the valueit
is pointing at. This dictionary is simply axiomatized as a module
calledPM declaring a polymorphic abstract typet:

Module PM.
Parameter t : Set → Set.

ThusPM.t a is the type of a dictionary for references of typea.
Three operations are provided on this type:

Parameter find : ∀a, t a → pointer → option a.
Parameter add : ∀a, t a → pointer → a → t a.
Parameter new : ∀a, t a → pointer.

find returns the value possibly associated to a pointer (that isNone
if there is no associated value andSome v otherwise);add adds
a new mapping; and finallynew returns a fresh reference (that is
a reference which is not yet mapped to any value). Three axioms
describe the behavior of these three operations:

Axiom find_add_eq :
∀a, ∀m:t a, ∀p:pointer, ∀v:a,
find (add m p v) p = Some v.

Axiom find_add_neq :
∀a, ∀m:t a, ∀p p′:pointer, ∀v:a,
~p′=p → find (add m p v) p′ = find m p′.

Axiom find_new :
∀a, ∀m:t a, find m (new m) = None.

End PM.

This axiomatization is obviously consistent, since we could realize
it using natural numbers for pointers and a finite map (e.g. an
association list) for the dictionary.

Then the heap can be viewed as a set of such dictionaries, since
ML typing prevents aliasing between references of different types
(we are not considering polymorphic references here). The next two
sections use this memory model to verify both persistent arrays and
persistent union-find data structures.

4.3 Verifying Persistent Arrays

The type of persistent arrays is directly modelled as thepointer
type for references. Thedata type is the same sum type as in the
OCAML code:

Inductive data : Set :=
| Arr : data
| Diff : Z → Z → pointer → data.

We use the fact that there is a single instance of theArr node to
make it a constant constructor. Indeed, we model the OCAML heap
by the pair made of a dictionary mapping pointers to values oftype
data and of the contents of the array designated byArr, which is
here modelled as a function fromZ to Z:

Record mem : Set := { ref : PM.t data; arr : Z→Z }.

It is clear that we only model the part of the heap relative to
a single persistent array and its successive versions, but this is
enough for this formal proof (since there is no operation taking
several persistent arrays as arguments, which would require a more
complex model).

As defined above, our model already includes much more possi-
ble situations than allowed by the solecreate andset operations
(exactly as the OCAML type from Section 2.3.1 does not exclude
a priori the construction of cyclic values for instance). To intro-
duce the structural invariant of persistent arrays, we introduce the
following inductive predicate,pa valid, which states that a given
reference is a valid persistent array:

Inductive pa_valid (m: mem) : pointer → Prop :=
| array_pa_valid :
∀p, PM.find (ref m) p = Some Arr →
pa_valid m p

| diff_pa_valid :
∀p i v p′,
PM.find (ref m) p = Some (Diff i v p′) →
pa_valid m p′ → pa_valid m p.

This definition says that the reference points either to a valueArr
or to a valueDiff i v p′ with p′ being itself a valid persistent array.
Note that it implies the absence of cycles due to its inductive nature.

To express the specifications of theget andset functions, we
relate each persistent array to the function over{0, 1, . . . , n − 1}
that it represents. Such a function is directly modelled as avalue of
typeZ→Z. We introduce the following inductive predicate relating
a reference to a function:

Inductive pa_model (m: mem)
: pointer → (Z → Z) → Prop :=
| pa_model_array :
∀p, PM.find (ref m) p = Some Arr →
pa_model m p (arr m)

| pa_model_diff :



∀p i v p′,
PM.find (ref m) p = Some (Diff i v p′) →
∀f, pa_model m p′ f →
pa_model m p (upd f i v).

whereupd is the pointwise update of a function, defined as

Definition upd (f:Z→Z) (i:Z) (v:Z) :=
fun j ⇒ if Z_eq_dec j i then v else f j.

As we can notice, the definition ofpa model is analogous to the
one ofpa valid. It is even clear thatpa valid m p holds as soon
aspa model m p f does for any functionf . But it is convenient to
distinguish the two notions, as we will see in the following.

We can now give specifications to theget andset functions.
Opting for a definition-as-proof, a possible type forget is:

Definition get :
∀m, ∀p, pa_valid m p →
∀i, { v:Z | ∀f, pa_model m p f → v = f i }.

The precondition states that the referencep must designate a valid
persistent array and the postcondition states that the returned value
v must bef v for any functionf modelled byp through the
memory layoutm. The correctness proof ofget does not seem
much difficult, due to the tautological nature of its specification.
Though it raises a serious issue, namely its termination. Indeed,
get only terminates because we assumedp to be a valid persistent
array. If not, we could have a circularity on the heap (such as
the one resulting fromlet rec p = ref (Diff (0,0,p))) and
some calls toget would not terminate. To express the termination
property,i.e. that referencep eventually leads to theArr node, we
introduce the predicatedist m p n which states that the distance
from p to theArr node isn. Then we can introduce the relation
R m, for a given memorym, stating that a pointer is closer to the
Arr node than another one:

R m p1 p2 ≡ ∃n1, ∃n2, distm p1 n1∧distm p2 n2∧n1 < n2

Thenget can be defined using a well-founded induction over this
relation. Since it is easy to show that ifp points toDiff i v p′

then R m p′ p holds, under the hypothesispa valid m p, the
termination ofget follows.

The set function has a more complex specification since it
must not only express the assignment but also the persistence of
the initial argument. A possible specification is the following:

Definition set :
∀m:mem, ∀p:pointer, ∀i v:Z,
pa_valid m p →
{ p′:pointer & { m′:mem |
∀f, pa_model m p f →

pa_model m′ p f ∧
pa_model m′ p′ (upd f i v) } }.

Here the function is returning both the resulting referencep′ but
also the new memory layoutm′. The precondition is still the va-
lidity of the persistent array passed as argument. The postcondition
states that for any functionf that p was modelling in the initial
memorym thenp keeps modellingf in the new memorym′ and
p′ is modelling the result of the assignment, that is the function
upd f i v.

The proof of correctness forset only contains a single dif-
ficulty: we need to show that the allocation of a new reference
(through the functionPM.new) does not modify the allocated values
on the heap. This is expressed by the following separation lemma:

Lemma pa_model_sep :
∀m, ∀p, ∀d, ∀f,
pa_model m p f→

pa_model
(Build_mem

(PM.add (ref m) (PM.new (ref m)) d)
(arr m))

p f.

The proofs of correctness forget andset represent a total of 140
lines of Coq script, including all auxiliary lemmas and definitions.
This proof does not include the verification of functionreroot
(which is only an optimization). In our formalization, thisfunction
would have the following type:

Definition reroot :
∀m:mem, ∀p:pointer, pa_valid m p →
{ m′:mem |
∀f, pa_model m p f → pa_model m′ p f }.

4.4 Verifying Persistent Union-Find

To model the union-find data structure, we set the number of el-
ements once and for all, as a parameterN of typeZ. We can omit
the management of ranks without loss of generality, since they only
account for the complexity, not for the correctness. Thus the union-
find data structure reduces to a single array (theparent array). We
can reuse the previous memory model and we can model a union-
find data structure as a value of typepointer.

As we did for persistent arrays, we first define a validity predi-
cate. Indeed, a union-find data structure is not made of any persis-
tent array but of one modelling a function mapping each integer in
[0, N − 1] to a representative, in one or several steps. In particu-
lar, this property will ensure the termination. To define this validity
predicate, we introduce the notion of representatives, fora function
f modelling the persistent arrayparent, as a relationrepr f i j
meaning that the representative ofi is j:

Inductive repr (f: Z→Z) : Z→Z→Prop :=
| repr_zero :

∀i, f i = i → repr f i i
| repr_succ :

∀i j r, f i = j → 0≤j<N →
~j=i → repr f j r → repr f i r.

Then we can define the validity notion as both the validity of the
persistent array and the existence of a representative for any integer
in [0, N − 1]:

Definition reprf (f:Z→Z) :=
(∀i, 0≤i<N → 0≤f i<N) ∧
(∀i, 0≤i<N → ∃j, repr f i j).

Definition uf_valid (m:mem) (p:pointer) :=
pa_valid m p ∧
∀f, pa_model m p f → reprf f.

Specifying thefind function is more complex than simply
saying that the returned value is the representative. Indeed, the
find function modifies the memory, due to path compression,
and thus we need to express the invariance of classes throughout
this compression. We first define the property for two functions to
define the same set of representatives:

Definition same_reprs f1 f2 :=
∀i, 0≤i<N → ∀j, repr f1 i j ↔ repr f2 i j.

Then we can specify thefind function (actually thefind aux
function returning the new persistent array, modelled hereas a
pointerp′):

Definition find :
∀m, ∀p, uf_valid m p →
∀x, 0≤x<N →
{ r:Z & { p′:pointer & { m′:mem |



uf_valid m′ p′ ∧
∀f, pa_model m p f → repr f x r ∧
∀f ′, pa_model m′ p′ f ′ → same_reprs f f ′ } } }.

Once again the proof requires a well-founded induction, here on
the distance fromx to its representative.

To specify theunion function, we need a notion of equivalent
elements,i.e.belonging to the same class:

Definition equiv f x y :=
∀cx cy, repr f x cx → repr f y cy → cx=cy.

Then we can specifyunion (the assignmenth.parent <- f is
here modelled by the returned pointerp′):

Definition union :
∀m, ∀p, uf_valid m p →
∀x y, 0≤x<N → 0≤y<N →
{ p′:pointer & { p1:pointer & { m′:mem |

uf_valid m′ p1 ∧ uf_valid m′ p′ ∧
∀f1, pa_model m p f1 →
((∀f2, pa_model m′ p1 f2 →

same_reprs f1 f2)
∧
(∀f ′, pa_model m′ p′ f ′ →
∀a b, 0≤a<N → 0≤b<N →

(equiv f ′ a b ↔
(equiv f1 a b ∨
(equiv f1 a x ∧ equiv f1 b y) ∨
(equiv f1 b x ∧ equiv f1 a y))))))

}}}.

The slightly complex postcondition states several properties: first
the persistence of the initial structure (it is still valid in the new
memory and defines the same set of representatives); then theva-
lidity of the new structure; finally the behavioral property, namely
thatunion indeed merges the two classes ofx andy. We express
this last property by saying that, for any elementsa andb, a and
b are in the same class after the merge (functionf ′) if and only if
they were already in the same class (functionf1) or a andb were
both in the initial classes ofx andy.

The proofs of correctness offind andunion represent a total
of 600 lines of Coq script. Note that, since we omitted the man-
agement of ranks,union a b always appends the class ofb as a
subtree of the representative ofa. To be fully correct, we should
also consider the converse as ifunion was performing a nondeter-
ministic choice.

5. Conclusion
We have presented a persistent data structure for the union-find
problem which is as efficient as the imperative Tarjan’s algo-
rithm [17] on realistic benchmarks. In particular, our solution is
exactly the same as the imperative one when used linearlyi.e.
without any backtracking.

Our solution is built from a persistent version of Tarjan’s algo-
rithm together with an efficient implementation of persistent arrays
following an idea from Baker [4, 5]. Though persistent, these two
data structures make heavy use of side effects. Contrary to awidely
spread idea, persistent data structures are not necessarily purely ap-
plicative (even if excellent books such as Okasaki’s [14] only focus
on purely applicative solutions). As a consequence, it is less ob-
vious to convince oneself of the correctness of the implementation
and this is why we also conducted a formal verification of our code.

Another consequence of the imperative nature of this persistent
data structure is that it is not thread-safe. Indeed, the assignment
in function union is atomic but assigments in functionset are

not (functionreroot makes a lot of assigments all over the data
structure).

The most efficient version we finally obtained actually uses ar-
rays that are not fully persistent. Indeed, they must only beused to
come back to previous versions (which is the typical use of persis-
tent structures in a backtracking algorithm). Thissemi-persistence
currently has a dynamic nature (the data is made invalid whenwe
do the backtrack) but it would be even more efficient if we checked
statically the legal use of this semi-persistence. Such a static anal-
ysis is work in progress [9].

A. Imperative Union-Find Algorithm
The following code implements the optimal imperative solution for
union-find, as described in the introduction. The data structure is a
record containing two arrays:

type t = { parent : int array; rank : int array }

parent links together the elements of each class andrank con-
tains the size of each class. Creation is straightforward, using
Array.init to create an array where each element is mapped
to itself:

let create n =
{ parent = Array.init n (fun i → i);

rank = Array.create n 0 }

The functionfind recursively follows the links until it finds the
representative (that is an element mapped to itself), performing path
compression along the way:

let rec find uf i =
let pi = uf.parent.(i) in
if pi == i then

i
else begin

let ci = find uf pi in
uf.parent.(i) ← ci; (* path compression *)
ci

end

Finally, union maps the representative of the smallest class to the
one of the largest and updates the rank when necessary:

let union ({ parent = p; rank = r } as uf) x y =
let cx = find uf x in
let cy = find uf y in
if cx != cy then begin

if r.(cx) > r.(cy) then
p.(cy) ← cx

else if r.(cx) < r.(cy) then
p.(cx) ← cy

else begin
r.(cx) ← r.(cx) + 1;
p.(cy) ← cx

end
end

